

Constructive Formalization of Regular Languages

Base revision 795b873, Tue Oct 9 00:09:30 2012 +0200, Jan-Oliver Kaiser.

Jan-Oliver Kaiser

October 9, 2012

Abstract

Our goal is to give a concise formalization of the equivalence between regular expressions, finite automata and the Myhill-Nerode characterization. We give procedures to convert between these characterizations and prove their correctness. Our development is done in the proof assistant Coq. We make use of the SSREFLECT plugin which provides support for finite types and other useful infrastructure for our purpose.

Contents

1	Introduction	4
1.1	Related work	4
1.2	Contributions	5
1.3	Outline	5
2	Coq and SSReflect	7
2.1	Coq	7
2.1.1	Non-structurally Recursive Functions	7
2.2	SSREFLECT	8
2.2.1	Finite Types	8
2.2.2	Boolean Reflection	8
2.2.3	Boolean Predicates	8
2.2.4	Equality	9
3	Decidable Languages	10
3.1	Definitions	10
3.1.1	Operations on Languages	11
3.2	Regular Languages	13
3.2.1	Regular Expressions	13
4	Finite Automata	15
4.1	Definition	15
4.1.1	Non-Deterministic Finite Automata	15
4.1.2	Deterministic Finite Automata	16
4.2	Connected Components	19
4.3	Emptiness	20
4.4	Deciding Equivalence of Finite Automata	21
4.5	Regular Expressions and Finite Automata	22
4.5.1	Regular Expressions to Finite Automata	22
4.5.2	Deciding Equivalence of Regular Expressions	26
4.5.3	Finite Automata to Regular Expressions	26

5	Myhill-Nerode	33
5.1	Definitions	33
5.1.1	Myhill Relations	34
5.1.2	Nerode Relations	35
5.1.3	Myhill-Nerode Theorem	35
5.2	Finite Automata, Myhill relations, and Nerode relations . . .	36
5.2.1	Finite Automata to Myhill relations	36
5.2.2	Myhill relations to weak Nerode relations	37
5.2.3	Nerode relations to Finite Automata	38
5.3	Minimizing Equivalence Classes	39
6	Conclusion	44

Chapter 1

Introduction

Our goal is to give a concise formalization of the equivalence between regular expressions, finite automata and the Myhill-Nerode characterizations. We give procedures to convert between these characterizations and prove their correctness. Our development is done in the proof assistant COQ. We make use of the SSREFLECT plugin which provides support for finite types and other useful infrastructure for our purpose.

Regular languages are a well-studied class of formal languages. In their current form, they were first studied by Kleene [22], who introduced regular expressions. The concept of deterministic finite automata was introduced before Kleene's invention of regular expressions by Huffman [19] and Moore [27]. Rabin and Scott later introduced the concept non-deterministic finite automata [30], for which they were given the Turing award [4].

One of the challenges in the formalization was to find a suitable representation of quotient types in COQ, which has no notion of quotient types.

1.1 Related work

There have been many publications on formalizations of the theory of regular languages in recent years. Most of them investigate decidability of equivalence of regular expressions, often with a focus on automatically deciding Kleene algebras.

Coquand and Siles develop a decision procedure for equivalence of regular expressions [13] on the basis of Brzozowski derivatives [12] in COQ (with SSREFLECT) with the goal of providing a potentially executable tactic on top of the decision procedure. Their development weighs in at 7,500 lines of code, 700 of which serve as the basis of our formalization.

Krauss and Nipkow give a decision procedure for equivalence of regular expressions in Isabelle/HOL [24]. Their development is very concise with just over 1,000 lines of code. Being interested only in a correct (and efficient) tactic for deciding equivalences, they did not prove completeness and

termination.

Another decision procedure for equivalence of regular expressions is developed by Braibant and Pous [10], with the goal of deciding Kleene algebras in COQ. Their formalization is based on matrices and weighs in at 19,000 lines of code. It encompasses finite automata, regular expressions and the Myhill-Nerode theorem.

Moreira, Pereira and Sousa give a decision procedure for equivalence of regular expressions in COQ[28]. Their development is based on Antimirov's partial derivatives of regular expressions [2] and contains a refutation step to speed up inequality checking. It consists of 19,000 lines of code.

Asperti formalize a decision procedure for equivalence of regular expressions [5] based on the notion of pointed regular expressions [6]. This development was done in the Matita proof assistant [7]. It weighs in at 3,400 lines of code.

There is also been a paper by Wu, Zhang and Urban on formalizing the Myhill-Nerode theorem using only regular expressions and not, as is commonly done, finite automata [32]. The authors state that this unusual choice stems, at least partly, from the restrictions of Isabelle/HOL (and similar HOL-based theorem provers). In particular, the fact that Isabelle/HOL does not allow for quantification over types prevents straight-forward formalizations of finite automata. Their development consists of roughly 2,000 lines of code.

1.2 Contributions

Unlike recent publications in this area, we do not focus on executable decision procedures. Instead, our formalization is very close to the mathematical definitions given in [23]. Our development shows that COQ (particularly with SSREFLECT) is well suited for this kind of formalization. Furthermore, we have also developed a new characterization derived from the Nerode relation and proven it equivalent to all other characterizations. Our development weighs in at about 3,500 lines of code.

1.3 Outline

In Chapter 2 we give a brief introduction to COQ and SSREFLECT and introduce concepts that are relevant to our formalization.

In Chapter 3 we give basic definitions (words, languages, etc.). We also introduce decidable languages, regular languages and regular expressions. Furthermore, we prove the decidability of regular languages.

In Chapter 4 we introduce finite automata. We prove the equivalence of deterministic and non-deterministic finite automata. We also give a procedure to remove unreachable states from deterministic finite automata.

Furthermore, we prove decidability of emptiness and equivalence of finite automata. Finally, we prove that regular expressions and finite automata are equally expressive.

In Chapter 5 we introduce the Myhill-Nerode theorem. We give three different characterizations of regular languages based on the Myhill-Nerode theorem and prove them all equally expressive to finite automata.

Chapter 2

Coq and SSReflect

We decided to employ the Small Scale Reflection Extension (**SSReflect**¹) for the **Coq**² proof assistant. The most important factors in this decision were SSREFLECT's excellent support for finite types, list operations and graphs. SSREFLECT also introduces an alternative scripting language that can often be used to shorten the bookkeeping overhead of proofs considerably.

2.1 Coq

The Coq system is designed to develop mathematical proofs, and especially to write formal specifications, programs and to verify that programs are correct with respect to their specification. It provides a specification language named Gallina. Terms of Gallina can represent programs as well as properties of these programs and proofs of these properties. Using the so-called Curry-Howard isomorphism, programs, properties and proofs are formalized in the same language called Calculus of Inductive Constructions, that is a λ -calculus with a rich type system. All logical judgments in Coq are typing judgments. The very heart of the Coq system is the type-checking algorithm that checks the correctness of proofs, in other words that checks that a program complies to its specification. Coq also provides an interactive proof assistant to build proofs using specific programs called tactics [26].

Description,
citation

2.1.1 Non-structurally Recursive Functions

COQ allows us to define functions that do not recurse in a structural manner. We make use of COQ's **Function** syntax to define functions whose termination is proven by showing that a specified measure decreases in every recursive

¹<http://www.msr-inria.inria.fr/Projects/math-components>

²<http://coq.inria.fr/>

call. In our case, we will use the size of a finite set as the decreasing measure. When defining such a recursive functions, we have to prove that every recursive call reduces the size of the finite set.

2.2 SSReflect

SSREFLECT is a set of extensions to the proof scripting language of the Coq proof assistant. They were originally developed to support small-scale reflection. However, most of them are of quite general nature and improve the functionality of Coq in most basic areas such as script layout and structuring, proof context management and rewriting [16].

2.2.1 Finite Types

The most important feature of SSREFLECT for our purpose are finite types. Finite types are types that have a finite number of inhabitants. Their implementation is based on lists. Every element of a finite type is contained in the associated (de-duplicated) list and vice versa. SSREFLECT's support for finite types is based on canonical structures, instances of which come predefined for basic finite types and type constructors. This allows us to easily combine basic finite types such as `bool` with type constructors such as `option` and `sum`.

SSREFLECT provides boolean versions of the universal and existential quantifiers on finite types, `forallb` and `existsb`. We can compute the number of elements in a finite type `F` with `#|F|`. `enum` gives a list of all items of a finite type.

We can also create finite types from lists. Instances of these finite types can be specified with the `SeqSub` constructor, which takes as argument an element of the list and a proof that this element is contained in the list.

2.2.2 Boolean Reflection

SSREFLECT offers boolean reflections for decidable propositions. This allows us to switch back and forth between equivalent boolean and propositional predicates.

2.2.3 Boolean Predicates

SSREFLECT has special type for boolean predicates, `pred T := T -> bool`, where `T` is a type. We make use of SSREFLECT's syntax to specify boolean predicates. This allows us to specify predicates in a way that resembles set-theoretic notation, e.g. `[pred x | <boolean expression in x>]`. Furthermore, we can use the functions `pred1` and `pred0` to specify the singleton predicate and the empty predicate, respectively. The complement of a predicate can be

written as `[predC p]`. The syntax for combining predicates is `[pred? p1 & p2]`, with `?` being one of `U` (union), `I` (intersection) or `D` (difference). For predicates given in such a way, we write `y \in p` to express that `y` fulfills `p`. There is also syntax for the preimage of a predicate under a function which can be written as `[preim f of p]`.

There are also applicative (functional) versions of `predC`, `predU`, `predI`, `predD`, which are functions that take predicates as arguments and return predicates.

2.2.4 Equality

We can use `f =1 g` to express that the functions `f` and `g` agree in all arguments. If we regard `f` and `g` as sets, we can write `f =i g`, which is defined as `forall x, x \in f = x \in g`. COQ's equality `=` is intensional, which means that even if we have `f =1 g`, we will not, in general, be able to proof `f = g`. Thus, we will use `=1` and `=i` in COQ, when we write `=` mathematically. This expresses the notion of extensional equality that is usually assumed mathematically.

Chapter 3

Decidable Languages

We give basic definitions for languages, operators on languages and, finally, regular languages. We provide the corresponding formalizations from our development and prove their correctness.

3.1 Definitions

We closely follow the definitions from [18]. An **alphabet** Σ is a finite set of symbols. A **word** w is a finite sequence of symbols chosen from some alphabet. We use $|w|$ to denote the **length** of a word w . ε denotes the empty word. Given two words $w_1 = a_1 \cdots a_n$ and $w_2 = b_1 \cdots b_m$, the **concatenation** of w_1 and w_2 is defined as $a_1 \cdots a_n b_1 \cdots b_m$ and denoted $w_1 \cdot w_2$ or just $w_1 w_2$. A **language** is a set of words. The **residual language** of a language L with respect to symbol a is the set of words u such that au is in L . The residual is denoted $res_a(L)$. We define Σ^k to be the **set of words of length k**. The **set of all words** over an alphabet Σ is denoted Σ^* , i.e., $\Sigma^* = \bigcup_{k \in \mathbb{N}} \Sigma^k$. A language L is **decidable** if and only if there exists a boolean predicate that decides membership in L . We will only deal with decidable languages from here on. Throughout the remaining document, we will assume a fixed alphabet Σ .

We employ finite types to formalize alphabets. In most definitions, alphabets will not be made explicit. However, the same name and type will be used throughout the entire development. Words are formalized as sequences over the alphabet. Decidable languages are represented by functions from *word* to *bool*.

Variable char: finType.

Definition word := seq char.

Definition language := pred word.

Definition residual x L : language := [preim cons x of L].

3.1.1 Operations on Languages

We will later introduce a subset of the decidable language that is based on the following operations. For every operator, we will prove the decidability of the resulting language.

The **concatenation** of two languages L_1 and L_2 is denoted $L_1 \cdot L_2$ and is defined as the set of words $w = w_1w_2$ such that w_1 is in L_1 and w_2 is in L_2 . The **Kleene closure** of a language L is denoted L^* and is defined as the set of words $w = w_1 \cdots w_k$ such that $w_1 \dots w_k$ are in L . Note that $\varepsilon \in L^*$ ($k = 0$). We define the **complement** of a language L as $L \setminus \Sigma^*$, which we write as $\neg L$. Furthermore, we make use of the standard set operations **union** and **intersection**.

For our COQ development, we take Coquand and Siles's [13] implementation of these operators. `plus` and `prod` refer to union and intersection, respectively. Additionally, we also introduce the singleton languages (`atom`), the empty language (`void`) and the language containing only the empty word (`eps`).

Definition `conc L1 L2 : language :=`

`fun v => [exists i : 'L (size v).+1, L1 (take i v) && L2 (drop i v)].`

Definition `star L : language :=`

`fix star v := if v is x :: v' then conc (residual x L) star v' else true.`

Definition `compl L : language := predC L.`

Definition `plus L1 L2 : language := [predU L1 & L2].`

Definition `prod L1 L2 : language := [predI L1 & L2].`

Definition `atom x : language := pred1 [:: x].`

Definition `void : language := pred0.`

Definition `eps : language := pred1 [::].`

The definition of `conc` is based on a characteristic property of the concatenation of two languages. The following lemma proves this property.

Lemma 3.1.1. *Let $L_1, L_2, w = a_1 \cdots a_k$ be given. We have that*

$$w \in L_1 \cdot L_2 \iff \exists n \in \mathbb{N}. 0 < n \leq k \wedge a_1 \cdots a_{n-1} \in L_1 \wedge a_n \cdots a_k \in L_2.$$

Proof. “ \Rightarrow ” From $w \in L_1 \cdot L_2$ we have w_1, w_2 such that $w = w_1w_2 \wedge w_1 \in L_1 \wedge w_2 \in L_2$. We choose $n := |w_1| + 1$. We then have that $a_1 \cdots a_{n-1} = a_1 \cdots a_{|w_1|} = w_1$ and $w_1 \in L_1$ by assumption. Similarly, $a_n \cdots a_k = a_{|w_1|+1} \cdots a_k = w_2$ and $w_2 \in L_2$ by assumption.

“ \Leftarrow ” We choose $w_1 := a_1 \cdots a_{n-1}$ and $w_2 := a_n \cdots a_k$. By assumption we have that $w = w_1w_2$. We also have that $a_1 \cdots a_{n-1} \in L_1$ and $a_n \cdots a_k \in L_2$. It follows that $w_1 \in L_1$ and $w_2 \in L_2$. \square

Lemma `concP : forall {L1 L2 v},`

`reflect (exists2 v1, v1 \in L1 & exists2 v2, v2 \in L2 & v = v1 ++ v2)`
`(v \in conc L1 L2).`

The implementation of `star` makes use of a property of the Kleene closure, which is that any nonempty word in L^* can be seen as the concatenation of a nonempty word in L and a (possibly empty) word in L^* . This property allows us to implement `star` as a structurally recursive predicate. The following lemma proves the correctness of this property.

Lemma 3.1.2. *Let $L, w = a_1 \cdots a_k$ be given. We have that*

$$w \in L^* \iff \begin{cases} a_2 \cdots a_k \in \text{res}_{a_1}(L) \cdot L^*, & \text{if } |w| > 0; \\ w = \varepsilon, & \text{otherwise.} \end{cases}$$

Proof. “ \Rightarrow ” We do a case distinction on $|w| = 0$.

1. $|w| = 0$. It follows that $w = \varepsilon$.
2. $|w| \neq 0$, i.e. $|w| > 0$. From $w \in L^*$ we have $w = w_1 \cdots w_l$ such that $w_1 \cdots w_l$ are in L . There exists a minimal n such that $|w_n| > 0$ and for all $m < n$, $|w_m| = 0$. Let $w_n = b_1 \cdots b_p$. We have that $b_2 \cdots b_p \in \text{res}_{b_1}(L)$. Furthermore, we have that $w_{n+1} \cdots w_l \in L^*$. We also have $a_1 = b_1$ and $w = a_1 \cdots a_k = w_n \cdots w_l$. Therefore, we have $a_2 \cdots a_k \in \text{res}_{a_1}(L) \cdot L^*$.

“ \Leftarrow ” We do a case distinction on the disjunction.

1. $w = \varepsilon$. Then $w \in L^*$ by definition.
2. $a_2 \cdots a_k \in \text{res}_{a_1}(L) \cdot L^*$. By Lemma 3.1.1 we have n such that $a_2 \cdots a_{n-1} \in \text{res}_{a_1}(L)$ and $a_n \cdots a_k \in L^*$. By definition of `res`, we have $a_1 \cdots a_{n-1} \in L$. Furthermore, we also have $a_n \cdots a_k = w_1 \cdots w_l$ such that $w_1 \cdots w_l$ are in L . We choose $w_0 := a_1 \cdots a_{n-1}$. It follows that $w = w_0 w_1 \cdots w_l$ with w_0, w_1, \dots, w_l in L . Therefore, $w \in L^*$.

□

The formalization of Lemma 3.1.2 connects the formalization of `star` to the mathematical definition. The propositional formula given here appears slightly more restrictive than our mathematical definition as it requires all words from L to be nonempty. Mathematically, however, this is no restriction.

Listing 3.1: Formalization of Lemma 3.1.2

Lemma `starP` : `forall {L v},`
`reflect (exists2 vv, all [predD L & eps] vv & v = flatten vv)`
`(v \in star L).`

Theorem 3.1.3. *The decidable languages are closed under concatenation, Kleene star, union, intersection and complement.*

Proof. We have already give algorithms for all operators. It remains to show that they are correct. For concatenation and the Kleene star, we have shown in Lemma 3.1.1 and Lemma 3.1.2 that the formalizations are equivalent to the mathematical definitions. The remaining operators (union, intersection, complement) can be applied directly to the result of the languages' boolean decision functions. \square

3.2 Regular Languages

Definition 3.2.1. *The set of regular languages REG is defined to be exactly those languages generated by the following inductive definition:*

$$\begin{array}{c} \overline{\emptyset \in REG} \qquad \overline{\{\varepsilon\} \in REG} \qquad \frac{a \in \Sigma}{\{a\} \in REG} \qquad \frac{L \in REG}{L^* \in REG} \\[10pt] \frac{L_1 \in REG \quad L_2 \in REG}{L_1 \cup L_2 \in REG} \qquad \frac{L_1 \in REG \quad L_2 \in REG}{L_1 \cdot L_2 \in REG} \end{array}$$

3.2.1 Regular Expressions

Regular expressions mirror the definition of regular languages very closely.

Definition 3.2.2. *We will consider **extended regular expressions** that include negation (Not), intersection (And) and a single-symbol wildcard (Dot). Therefore, we have the following syntax for regular expressions:*

$$r, s := \emptyset \mid \varepsilon \mid . \mid a \mid r^* \mid r + s \mid r \& s \mid rs \mid \neg r$$

The language of an extended regular expression is defined as follows:

$$\begin{array}{ll} \mathcal{L}(\emptyset) = \emptyset & \mathcal{L}(r^*) = \mathcal{L}(r)^* \\ \mathcal{L}(\varepsilon) = \{\varepsilon\} & \mathcal{L}(r + s) = \mathcal{L}(r) \cup \mathcal{L}(s) \\ \mathcal{L}(\cdot) = \Sigma & \mathcal{L}(r \& s) = \mathcal{L}(r) \cap \mathcal{L}(s) \\ \mathcal{L}(a) = \{a\} & \mathcal{L}(rs) = \mathcal{L}(r) \cdot \mathcal{L}(s) \end{array}$$

Definition 3.2.3. *We say that two regular expressions r and s are equivalent if and only if*

$$\mathcal{L}(r) = \mathcal{L}(s).$$

We will later show that equivalence of regular expressions is decidable.

We take the implementation of regular expressions from Coquand and Siles's development ([13]), which is also based on SSREFLECT and comes with helpful infrastructure for our proofs. The semantics defined in Definition 3.2.2 can be given as a boolean function.

Listing 3.2: Regular Expressions

```

Inductive regular_expression :=
| Void
| Eps
| Dot
| Atom of symbol
| Star of regular_expression
| Plus of regular_expression & regular_expression
| And of regular_expression & regular_expression
| Conc of regular_expression & regular_expression
| Not of regular_expression .

Fixpoint mem_reg e :=
match e with
| Void => void
| Eps => eps
| Dot => dot
| Atom x => atom x
| Star e1 => star (mem_reg e1)
| Plus e1 e2 => plus (mem_reg e1) (mem_reg e2)
| And e1 e2 => prod (mem_reg e1) (mem_reg e2)
| Conc e1 e2 => conc (mem_reg e1) (mem_reg e2)
| Not e1 => compl (mem_reg e1)
end.

```

We will later prove that extended regular expressions are equivalent to the inductive definition of regular languages in 3.2.1. In order to do that, we introduce a predicate on regular expressions that distinguishes **standard regular expressions** from **extended regular expressions** (as introduced above). The grammar of standard regular expression is as follows:

$$r, s := \emptyset \mid \varepsilon \mid a \mid r^* \mid r + s \mid rs$$

Note that standard regular expressions correspond directly to Definition 3.2.1.

Connect
stan-
dard reg-
exp to
reg. lan-
guages

Chapter 4

Finite Automata

Another way of characterizing regular languages are finite automata (FA)[18]. We will show that the languages of finite automata are exactly the regular languages. Furthermore, we will also derive a decision procedure for emptiness of an automaton's language. Based on that, we will give a decision procedure for equivalence of regular expressions.

4.1 Definition

A finite automaton consists of

1. finite set of states Q ,
2. a starting state $s \in Q$,
3. a set of final states $F \subseteq Q$
4. and a state-transition relation δ .

We define a **run** of a word $w \in \Sigma^*$ on an automaton $A = (Q, s, F, \delta)$ as a sequence of states σ such that for every two consecutive positions $i, i + 1$ in σ we have $(\sigma_i, w_i, \sigma_{i+1}) \in \delta$. A word w is **accepted** by A in state x if and only if there exists a run σ of w on A such that $\sigma_0 = x \wedge \sigma_{|\sigma|-1} \in F$. The resulting set of accepted words is denoted by $\mathcal{L}_x(A)$. The **language** of A is exactly $\mathcal{L}_s(A)$ and is denoted $\mathcal{L}(A)$.

4.1.1 Non-Deterministic Finite Automata

Finite automata can be **non-deterministic** (NFA) in the sense that there may exist multiple distinct runs for a word.

Listing 4.1: Non-Deterministic Finite Automata

```
Record nfa : Type :=  
{ nfa_state :> finType;
```



```

nfa_s : nfa_state ;
nfa_fin : pred nfa_state ;
nfa_step : nfa_state -> char -> pred nfa_state }.

```

```

Fixpoint nfa_accept (x: A) w :=
match w with
| [] => nfa_fin A x
| a :: w => [ exists y, (nfa_step A x a y) && nfa_accept y w ]
end.

```

Definition nfa_lang := [pred w | nfa_accept (nfa_s A) w].

The acceptance criterion given here avoids the matter of runs. In many cases, this will help us with proofs by induction on the accepted word. However, we will need runs in some of the proofs. Due to the fact that runs are not unique on NFAs, we give a predicate that decides if a sequence of states is a run on A for a word w . We then show that the acceptance criterion given above corresponds to the mathematical definition in terms of runs.

```

Fixpoint nfa_run x (xs : seq A) (w: word) {struct xs} :=
match xs, w with
| y :: xs', a :: w' => nfa_step A x a y && nfa_run y xs' w'
| [], [] => true
| -, - => false
end.

```

Lemma nfa_run_accept x w:
 reflect (exists2 xs, nfa_run x xs w & last x xs \in nfa_fin A)
 (nfa_accept x w).

4.1.2 Deterministic Finite Automata

For functional δ , we speak of **deterministic** finite automata (DFA). In this case, we write δ as a function in our COQ development.

Listing 4.2: Deterministic Finite Automata

```

Record dfa : Type :=
{ dfa_state :> finType;
  dfa_s : dfa_state ;
  dfa_fin : pred dfa_state ;
  dfa_step : dfa_state -> char -> dfa_state }.

Fixpoint dfa_accept x w :=
match w with
| [] => dfa_fin A x
| a :: w => dfa_accept (dfa_step A x a) w
end.

Definition dfa_lang := [pred w | dfa_accept (dfa_s A) w].

```

Again, we avoid runs in our formalization of the acceptance criterion in favor of a acceptance criterion that is easier to work with in proofs. In this case, however, we can give a function that computes the unique run of a word on A . This allows us to give an alternative acceptance criterion that is closer to the mathematical definition. We also prove that both criteria are equivalent.

Fixpoint $\text{dfa_run}' (x: A) (w: \text{word}) : \text{seq } A :=$
match w **with**
 | $[:]$ $\Rightarrow [::]$
 | $a :: w \Rightarrow (\text{dfa_step } A \times a) :: \text{dfa_run}' (\text{dfa_step } A \times a) w$
end.
Lemma $\text{dfa_run_accept } x w : \text{last } x (\text{dfa_run}' x w) \setminus \text{in } \text{dfa_fin } A = (w \setminus \text{in } \text{dfa_accept } x).$

Equivalence of Automata

Definition 4.1.1. We say that two automata are **equivalent** if and only if their languages are equal.

Equivalence between DFA and NFA

Deterministic and non-deterministic finite automata are equally expressive. One direction is trivial since every DFA can be seen as a NFA. We prove the other direction using the powerset construction.

Definition 4.1.2. Given NFA A , we construct an equivalent DFA A_{det} in the following way:

$$\begin{aligned} Q_{det} &:= \{P \mid P \subseteq Q\} \\ s_{det} &:= \{s\} \\ F_{det} &:= \{P \in Q_{det} \mid P \cap F \neq \emptyset\} \\ \delta_{det} &:= \{(P, a, \bigcup_{p \in P} \{q \in Q \mid (p, a, q) \in \delta\}) \mid P \in Q_{det}, a \in \Sigma\}. \\ A_{det} &:= (Q_{det}, s_{det}, F_{det}, \delta_{det}). \end{aligned}$$

The formalization of A_{det} is straight-forward. The set of states is an implicit argument of the DFA constructor and thus not shown.

Definition $\text{nfa_to_dfa} :=$
 $\{ \mid \text{dfa_s} := \text{set1 } (\text{nfa_s } A);$
 $\text{dfa_fin} := [\text{pred } X: \{ \text{set } A \} \mid [\text{exists } x: A, (x \setminus \text{in } X) \&\& \text{nfa_fin } A \ x]];$
 $\text{dfa_step} := [\text{fun } X \ a \Rightarrow \bigcup_{x \mid x \setminus \text{in } X} \text{finset } (\text{nfa_step } A \ x \ a)] \mid \}$
 $\}$

Lemma 4.1.3. For all powerset states X and for all states x with $x \in X$ we have that

$$\mathcal{L}_x(A) \subseteq \mathcal{L}_X(A_{det}).$$

Proof. Let $w \in \mathcal{L}_x(A)$. We proof by induction on w that $w \in \mathcal{L}_X(A_{det})$.

- For $w = \varepsilon$ and $\varepsilon \in \mathcal{L}_x(A)$ we get $x \in F$ from $\varepsilon \in \mathcal{L}_x(A)$. From $x \in X$ we get $X \cap F \neq \emptyset$ and therefore $\varepsilon \in \mathcal{L}_X(A_{det})$.
- For $w = aw'$ and $aw' \in \mathcal{L}_x(A)$ we get y such that $w' \in \mathcal{L}_y(A)$ and $(x, a, y) \in \delta$. The latter gives us $y \in Y$ where Y is such that $(X, a, Y) \in \delta_{det}$. With $y \in Y$ and $w' \in \mathcal{L}_y(A)$ we get which gives us $w' \in \mathcal{L}_Y(A_{det})$ by inductive hypothesis. With $(X, a, Y) \in \delta_{det}$ we get $aw' \in \mathcal{L}_X(A_{det})$.

□

Lemma 4.1.4. *For all powerset states X and all words $w \in \mathcal{L}_X(A_{det})$ there exists a state x such that*

$$x \in X \wedge w \in \mathcal{L}_x(A).$$

Proof. By induction on w .

- For $w = \varepsilon$ and $\varepsilon \in \mathcal{L}_X(A_{det})$ we get $X \cap F \neq \emptyset$. Therefore, there exists x such that $x \in X$ and $x \in F$. Thus, we have $\varepsilon \in \mathcal{L}_x(A)$.
- For $w = aw'$ and $aw' \in \mathcal{L}_X(A_{det})$ we get Y such that $w' \in \mathcal{L}_Y(A_{det})$ and $(X, a, Y) \in \delta_{det}$. From the inductive hypothesis we get y such that $y \in Y$ and $w' \in \mathcal{L}_y(A)$. From $y \in Y$ and $(X, a, Y) \in \delta_{det}$ we get x such that $x \in X$ and $(x, a, y) \in \delta$. Thus, $aw' \in \mathcal{L}_x(A)$.

□

Theorem 4.1.5. *The powerset automaton A_{det} accepts the same language as A , i.e.*

$$\mathcal{L}(A) = \mathcal{L}(A_{det}).$$

Proof. “ \subseteq ” This follows directly from Lemma 4.1.3 with $x = s$ and $X = s_{det}$.

“ \supseteq ” From Lemma 4.1.4 with $X = s_{det}$ we get $\mathcal{L}_{s_{det}}(A_{det}) \subseteq \mathcal{L}_s(A)$, which proves the claim. □

The formalization of this proof is straight-forward and follows the plan laid out above. The corresponding Lemmas are:

Lemma `nfa_to_dfa_aux2` (x : A) w (X : nfa_to_dfa):

`x \in X \rightarrow nfa_accept A x w \rightarrow dfa_accept nfa_to_dfa X w.`

Lemma `nfa_to_dfa_aux1` (X : nfa_to_dfa) w:

`dfa_accept nfa_to_dfa X w \rightarrow [exists x, (x \in X) && nfa_accept A x w].`

Lemma `nfa_to_dfa_correct` : `nfa_lang A =i dfa_lang nfa_to_dfa .`

4.2 Connected Components

Finite automata can have isolated subsets of states that are not reachable from the starting state. Removing these states does not change the language. It will later be useful to have automata that only contain reachable states. Therefore, we define a procedure to extract the connected component containing the starting state from a given automaton.

Definition 4.2.1. Let $A = (Q, s, F, \delta)$ be a DFA. We define `reachable1` such that for all x and y , $(x, y) \in \text{reachable1} \iff \exists a, (x, a, y) \in \delta$. We define `reachable` $:= \{y \mid (s, y) \in \text{reachable1}^*\}$, where `reachable1*` denotes the transitive closure of `reachable1`. With this, we can define the connected automaton A_c :

$$\begin{aligned} Q_c &:= Q \cap \text{reachable} \\ s_c &:= s \\ F_c &:= F \cap \text{reachable} \\ \delta_c &:= \{(x, a, y) \mid (x, a, y) \in \delta \wedge x, y \in Q_c\} \\ A_c &:= (Q_c, s_c, F_c, \delta_c). \end{aligned}$$

We make use of SSREFLECT's *connect* predicate to extract a sequence of all states reachable from s . From this, we construct a finite type and use that as the new set of states. These new states carry a proof of reachability. We also have to give a transition function that ensures transitions always end in reachable states.

Definition `reachable1` $:=$ [fun $x \ y \Rightarrow$ [**exists** a , `dfa_step A1 x a == y`]].

Definition `reachable` $:=$ enum (connect `reachable1` (`dfa_s A1`)).

Lemma `reachable0` : `dfa_s A1 \in reachable`.

Lemma `reachable_step` $x \ a$: $x \in \text{reachable} \rightarrow \text{dfa_step } A1 \ x \ a \in \text{reachable}$.

Definition `dfa_connected` $:=$

```
{| dfa_s := SeqSub reachable0;
  dfa_fin := fun x => match x with SeqSub x _ => dfa_fin A1 x end;
  dfa_step := fun x a => match x with
    | SeqSub _ Hx => SeqSub (reachable_step _ a Hx)
  end |}.
```

Lemma 4.2.2. For every state $x \in \text{reachable}$ we have that

$$\mathcal{L}_x(A_c) = \mathcal{L}_x(A).$$

Proof. “ \subseteq ” Trivial. “ \supseteq ” We do an induction on w .

- For $w = \varepsilon$ we have $\varepsilon \in \mathcal{L}_x(A)$ and therefore $x \in F$. With $x \in \text{reachable}$ we get $x \in F_c$. Thus, $\varepsilon \in \mathcal{L}_x(A_c)$.

- For $w = aw'$ we have $y \in Q$ such that $(x, a, y) \in \delta$ and $w' \in \mathcal{L}_y(A)$. From $x \in \text{reachable}$ we get $y \in \text{reachable}$ by transitivity. Therefore, $(x, a, y) \in \delta_c$. The inductive hypothesis gives us $w' \in \mathcal{L}_y(A_c)$. Thus, $aw' \in \mathcal{L}_x(A_c)$.

□

Theorem 4.2.3. *The language of the connected automaton A_c is identical to that of the original automaton A , i.e.*

$$\mathcal{L}(A) = \mathcal{L}(A_c).$$

Proof. By reflexivity, we have $s \in \text{reachable}$. We use Lemma 4.2.2 with $x = s$ to prove the claim. □

The formalization of Lemma 4.2.2 and Theorem 4.2.3 is straight-forward.

Lemma `dfa_connected_correct' x (Hx: x \in reachable) :`
`dfa_accept dfa_connected (SeqSub Hx) =i dfa_accept A1 x.`

Lemma `dfa_connected_correct: dfa_lang dfa_connected =i dfa_lang A1.`

To make use of the fact that A_c is fully connected, we will prove a characteristic property of A_c . We will need this property of A_c in Chapter 5.

Definition 4.2.4. *A **representative** of a state x is a word w such that the unique run of w on A_c ends in x .*

Lemma 4.2.5. *There is a representative for every state $x \in Q_c$.*

Proof. x carries a proof of reachability. From this, we get a path through the graph of `reachable1` that ends in x . We build the representative by extracting the edges of the path and building a word from those. □

Lemma `dfa_connected_repr x :`
`exists w, last (dfa_s dfa_connected) (dfa_run dfa_connected w) = x.`

4.3 Emptiness

Given an automaton A , we can check if $\mathcal{L}(A) = \emptyset$. We simply obtain the connected automaton of A and check if there are any final states left.

Theorem 4.3.1. *The language of the connected automaton A_c is empty if and only if its set of final states F_c is empty, i.e.*

$$\mathcal{L}(A) = \emptyset \iff F_c = \emptyset.$$

Proof. By Theorem 4.2.3 we have $\mathcal{L}(A) = \mathcal{L}(A_c)$. Therefore, it suffices to show

$$\mathcal{L}(A_c) = \emptyset \iff F_c = \emptyset.$$

“ \Rightarrow ” We have $\mathcal{L}(A_c) = \emptyset$ and have to show that for all $x \in Q_c$, $x \notin F_c$. Let $x \in Q_c$. By Lemma 4.2.5 we get w such that the unique run of w on A_c ends in x . We use $\mathcal{L}(A_c) = \emptyset$ to get $w \notin \mathcal{L}(A_c)$, which implies that the run of w on A_c ends in a non-final state. By substituting the last state of the run by x we get $x \notin F_c$.

“ \Leftarrow ” We have $F_c = \emptyset$ and have to show that for all words w , $w \notin \mathcal{L}(A_c)$. We use $F_c = \emptyset$ to show that the last state of the run of w on A_c is non-final. Thus, $w \notin \mathcal{L}(A_c)$. □

Thus, emptiness is decidable.

Definition `dfa_lang_empty` := `#|dfa_fin dfa_connected| == 0`.

Lemma `dfa_lang_empty_correct`:
`reflect (dfa_lang A1 == pred0)`
`dfa_lang_empty`.

4.4 Deciding Equivalence of Finite Automata

Given finite automata A_1 and A_2 , we construct DFA A such that the language of A is the symmetric difference of the languages of A_1 and A_2 , i.e.,

$$\mathcal{L}(A) := \mathcal{L}(A_1) \ominus \mathcal{L}(A_2) = \mathcal{L}(A_1) \cap \neg \mathcal{L}(A_2) \cup \mathcal{L}(A_2) \cap \neg \mathcal{L}(A_1).$$

Theorem 4.4.1. *The equivalence of A_1 and A_2 is decidable, i.e.*

$$\mathcal{L}(A_1) = \mathcal{L}(A_2) \text{ if and only if } \mathcal{L}(A) \text{ is empty.}$$

Proof. The correctness of this procedure follows from the properties of the symmetric difference operator, i.e.

$$\mathcal{L}(A_1) \ominus \mathcal{L}(A_2) = \emptyset \iff \mathcal{L}(A_1) = \mathcal{L}(A_2).$$

Thus, equivalence is decidable. □

Definition `dfa_sym_diff` :=
`dfa_disj (dfa_conj A1 (dfa_compl A2)) (dfa_conj A2 (dfa_compl A1))`.

Lemma `dfa_sym_diff_correct`:
`dfa_lang_empty dfa_sym_diff <-> dfa_lang A1 == dfa_lang A2`.

4.5 Regular Expressions and Finite Automata

We prove that there is a finite automaton for every extended regular expression and vice versa. In fact, we can give a standard regular expression for every finite automaton. With this, we will prove that extended regular expressions are equivalent to standard regular expressions, thereby proving closure under intersection and negation.

4.5.1 Regular Expressions to Finite Automata

We prove that there exists an equivalent automaton for every extended regular expression. The structure of this proof is given by the inductive nature of regular expressions.

Depending on the constructor of the regular expression, we will construct a corresponding operation on DFAs or NFAs. Void, Eps, Dot, Atom, Plus, And and Not are very easy to implement on DFAs, whereas Star and Conc lend themselves well to NFAs.

We show our implementation for Void, Not, and Conc. We also give a short overview of the automaton corresponding to Star.

Void

Definition 4.5.1. *We define an empty DFA with a single, non-accepting state, i.e.*

$$A_\emptyset := (\{t\}, t, \emptyset, \{(t, a, t) \mid a \in \Sigma\}).$$

Lemma 4.5.2. *The language of the empty DFA is empty, i.e.*

$$\mathcal{L}(E) = \emptyset.$$

Proof. A_\emptyset has no final states, i.e. no run can end in a final state. \square

Definition `dfa_void` :=

```
{| dfa_s := tt;
   dfa_fin := pred0;
   dfa_step := [fun x a => tt] |}.
```

Lemma `dfa_void_correct` $x\ w$: $\sim\sim$ `dfa_accept dfa_void` $x\ w$.

Not

Definition 4.5.3. *Given DFA $A = (Q, s, F, \delta)$, the complement automaton A_\neg is constructed by switching accepting and non-accepting states, i.e.*

$$A_\neg := (Q, s, Q \setminus F, \delta).$$

Lemma 4.5.4. *For every state $x \in Q$, we have that $w \in \Sigma^*$ is accepted in x by A_\neg if and only if it is not accepted in x by A , i.e. $\mathcal{L}_x(A_\neg) = \Sigma^* \setminus \mathcal{L}_x(A)$*

Proof. We do an induction on w . For $w = \varepsilon$ we have $\varepsilon \in \mathcal{L}_x(A_{\neg}) \iff \varepsilon \in \mathcal{L}_x(A)$ from $x \in F \iff x \notin Q \setminus F$. For $w = aw'$ we get $(y, a, x) \in \delta$. By inductive hypothesis, $w' \in \mathcal{L}_x(A_{\neg}) \iff w' \notin \mathcal{L}_x(A)$. Thus, $aw' \in \mathcal{L}_y(A_{\neg}) \iff aw' \notin \mathcal{L}_y(A)$. \square

Lemma 4.5.5. A_{\neg} accepts the complement language of A , i.e. $\mathcal{L}(A_{\neg}) = \Sigma^* \setminus \mathcal{L}(A)$.

Proof. This follows directly from Lemma 4.5.4 with $x = s$. \square

Conc

The most common approach to build the concatenation automaton is to connect the final states of the first automaton to the starting state of the second automaton by an ε -translation. We do not allow ε -transitions in our automata. The reason for this is that we do not want to lose the direct correspondence of the length of word to the length of its run on an automaton. Thus, in order to build the concatenation automaton, we duplicate all edges from the starting state of the second automaton and add them to all final states of the first automaton. Since the final states may already have edges with the same labels, we chose to implement this operation on NFAs.

Definition 4.5.6. Given two NFAs $A_1 = (Q_1, s_1, F_1, \delta_1)$ and $A_2 = (Q_2, s_2, F_2, \delta_2)$ we construct the concatenation automaton in the following way:

$$\begin{aligned} Q_{\text{Conc}} &:= Q_1 \cup Q_2 \\ s_{\text{Conc}} &:= s_1 \\ F_{\text{Conc}} &:= \begin{cases} F_2 & \text{if } s_2 \notin F_2 \\ F_2 \cup F_1 & \text{if } s_2 \in F_2 \end{cases} \\ \delta_{\text{Conc}} &:= \delta_1 \cup \delta_2 \cup \{(x, a, y) \mid x \in Q_1, y \in Q_2, (s_2, a, y) \in \delta_2\} \\ A_{\text{Conc}} &:= (Q_{\text{Conc}}, s_{\text{Conc}}, F_{\text{Conc}}, \delta_{\text{Conc}}). \end{aligned}$$

Definition `nfa_conc : nfa :=`

```
{| nfa_s := inl _ (nfa_s A1);
  nfa_fin := [fun x =>
    match x with
    | inl x => nfa_fin A1 x && nfa_fin A2 (nfa_s A2)
    | inr x => nfa_fin A2 x
    end];
  nfa_step := fun x a y =>
    match x,y with
    | inl x, inl y => nfa_step A1 x a y
    | inl x, inr y => nfa_fin A1 x && nfa_step A2 (nfa_s A2) a y
    | inr x, inr y => nfa_step A2 x a y
    | inr x, inl y => false
    end |}.
```


Before we prove the correctness of A_{Conc} , we need a number of auxiliary lemmas.

Lemma 4.5.7. *Every run of A_2 can be mapped to a run in A_{Conc} .*

Proof. Let σ be a run starting in x for $w \in \Sigma^*$ on A_2 . We do an induction on σ .

1. For $\sigma = x$ we have $w = \varepsilon$. Therefore, we have that σ is also a run starting in x for ε on A_{Conc} .
2. For $\sigma = xy\sigma'$ we have $w = aw'$, $(x, a, y) \in \delta_2$. By definition of δ_{Conc} we also have $(x, a, y) \in \delta_{Conc}$. By inductive hypothesis, we have that $y\sigma'$ is a run for w' starting in y on A_{Conc} . Thus, $xy\sigma'$ is a run for aw' starting in x on A_{Conc} .

□

Lemma `nfa_conc_cont x xs w:`

`nfa_run A2 x xs w`

`-> nfa_run nfa_conc (inr _ x) (map (@inr A1 A2) xs) w.`

The next lemma shows that, in A_{Conc} , the final states of A_1 have all transitions that the starting state of A_2 also has. Consequently, the accept the same words.

Lemma 4.5.8. *Let $x \in F_1$. Let $w \in \mathcal{L}(A_2)$. Then $w \in \mathcal{L}_x(A_{Conc})$.*

Proof. By induction on w .

1. For $w = \varepsilon$ we have get $s \in F_2$ by $\varepsilon \in \mathcal{L}(A_2)$ and, thus, $x \in F_{Conc}$ by definition.
2. For $w = aw'$ we have $y \in Q_2$ such that $(s, a, y) \in \delta_2$ and thus $(x, a, y) \in \delta_{Conc}$. We also have $w' \in \mathcal{L}_y(A_2)$ and thus, by Lemma 4.5.7, $w' \in \mathcal{L}_y(A_{Conc})$. Thus, $aw' \in \mathcal{L}_x(A_{Conc})$.

□

Lemma `nfa_conc_fin1 x1 w:`

`nfa_fin A1 x1 ->`

`nfa_lang A2 w ->`

`nfa_accept nfa_conc (inl _ x1) w.`

The following lemma is one direction of the proof of correctness of A_{Conc} .

Lemma 4.5.9. *Let $x \in Q_1$, $w_1 \in \mathcal{L}_x(A_1)$, and $w_2 \in \mathcal{L}(A_2)$. Then $w_1w_2 \in \mathcal{L}_x(A_{Conc})$.*

Proof. By induction on w_1 .

1. For $w_1 = \varepsilon$ we get $x \in F_1$ and thus, by Lemma 4.5.8, the claim follows.
2. For $w_1 = aw'_1$ we get $(x, a, y) \in \delta_1$ and thus $(x, a, y) \in \delta_{Conc}$. By inductive hypothesis, the claim follows.

□

Lemma `nfa_conc_aux2` $x \text{ w1 w2}$:

```
nfa_accept A1 x w1 ->
nfa_lang A2 w2 ->
nfa_accept nfa_conc (inl _ x) (w1 ++ w2).
```

The next lemma constitutes the other direction. Its statement is very general, even though we will only need one of the two cases for the proof of correctness of A_{Conc} . However, with the second case, there is no straightforward inductive proof.

Lemma 4.5.10. *Let $x \in Q_{Conc}$. Let $w \in \mathcal{L}_x(A_{Conc})$. Then, either*

$$x \in Q_1 \wedge \exists w_1. \exists w_2. w = w_1 w_2 \wedge w_1 \in \mathcal{L}_x(A_1) \wedge w_2 \in \mathcal{L}(A_2), \quad (*)$$

or

$$x \in Q_2 \wedge w \in \mathcal{L}_x(A_2). \quad (**)$$

Proof. By induction on w .

1. For $w = \varepsilon$ we get either $x \in F_1$ and $s_2 \in F_2$ or $x \in F_2$. In the first case, we need to prove (*), which we do by choosing $w_1 := \varepsilon$ and $w_2 := \varepsilon$. In the second case, we need to prove $\varepsilon \in \mathcal{L}_x(A_2)$ and thus $x \in F_2$ which we have by assumption.
2. For $w = aw'$ we get y such that $(x, a, y) \in \delta_{Conc}$ and $w' \in \mathcal{L}_y(A_{Conc})$. We are left with four cases, depending on the origin of x and y .
 - (a) For $x, y \in Q_2$ we have $(x, a, y) \in Q_2$ and claim (**) follows.
 - (b) For $x, y \in Q_1$ we prove (*). By inductive hypothesis we get w_1 and w_2 such that (*) holds for y .
 - (c) For $x \in Q_1$ and $y \in Q_2$ the claim follows with $w_1 := \varepsilon$ and $w_2 := aw'$ by inductive hypothesis.
 - (d) For $x \in Q_2$ and $y \in Q_1$ we have $(x, a, y) \in \delta_{Conc}$, which is a contradiction.

□

Lemma `nfa_conc_aux1` $X \text{ w}$:

```
nfa_accept nfa_conc X w ->
match X with
| inl x => exists w1, exists w2, (w == w1 ++ w2) && (nfa_accept A1 x w1) && nfa_lang A2 w2
| inr x => nfa_accept A2 x w
end.
```

Corollary 4.5.11. *The language of A_{Conc} is the concatenation of the languages of A_1 and A_2 , i.e. $\mathcal{L}(A_{\text{Conc}}) = \mathcal{L}(A_1) \cdot \mathcal{L}(A_2)$.*

Proof. Follows directly from Lemma 4.5.9 and Lemma 4.5.10. \square

Lemma `nfa_conc_correct`: `nfa_lang nfa_conc =i conc (nfa_lang A1) (nfa_lang A2)`.

Star

The most common construction for the star automaton works by adding the starting state to the set of final states and connecting all final states to the starting state by ε -transitions.

Again, our construction differs from this. First, we construct an automaton that accepts the Kleene closure of the language of the given automaton, excluding the empty word, which we call `nfa_repeat`. The reason for this is that we can easily construct this automaton much in the same way we constructed the concatenation automaton.

We duplicate all edges from the starting state and add them to the final states. The resulting automaton accepts the Kleene closure of the language of the given automaton, but not the empty word. Since we have already constructed an automaton that accepts the empty word, and a disjunction operation on automata, we simply combine those with our newly constructed automaton to form the star automaton.

Definition `nfa_star` := `(dfa_disj dfa_eps (nfa_to_dfa nfa_repeat))`.

Lemma `nfa_star_correct`: `dfa_lang nfa_star =i star (nfa_lang A1)`.

4.5.2 Deciding Equivalence of Regular Expressions

Based on our procedure to construct an equivalent automaton from a regular expression, we can decide equivalence of regular expressions. Given r_1 and r_2 , we construct equivalent DFA A_1 and A_2 as above.

4.5.3 Finite Automata to Regular Expressions

We prove that there is an equivalent standard regular expression for every finite automaton. There are three ways to prove this.

The first one is a method called “**state removal**” [11] (reformulated in [17]), which works by sequentially building up regular expressions on the edges between states. In every step, one state is removed and its adjacent states’ edges are updated to incorporate the missing state into the regular expression. Finally, only two states remain. The resulting edges can be combined to form a regular expression that recognizes the language of the initial automaton.

The second method is known as “**Brzozowski’s method**” [12] and builds upon Brzozowski derivatives of regular expressions. This method is algebraic in nature and arrives at a regular expression by solving a system of linear equations on regular expressions. Every state is assigned an unknown regular expression. The intuition of these unknown regular expressions is that they recognize the words accepted in their associated state. The system is solved by substitution and Arden’s lemma [3]. The regular expressions associated with the starting state recognizes the language of the automaton.

The third method, which we chose for our development, is due to Kleene [22]. It is known as the “**transitive closure method**”. This method recursively constructs a regular expression that is equivalent to the given automaton. For the remainder of the chapter, we assume that we are given a DFA $A = (Q, s, F, \delta)$.

The idea of the transitive closure method is that we can give a regular expression to describe the path between any two states x and y . This regular expression accepts every word whose run σ on A starts in x ends in y . In fact, we can even give such a regular expression if we limit the set of paths through which the run is allowed to pass. We will call this set X . Passing through, here, means that the restriction applies only to states that are traversed, i.e. not to the beginning or end of the run.

If we take X to be the empty set, we only consider two types of runs. First, if $x \neq y$, every transition from x to y constitutes one (singleton) word. Conversely, if there is a word which does not pass through a state and whose run on A starts in x and ends in y , it can only be a singleton word consisting of one of the transitions from x to y . Therefore, the corresponding regular expression is the disjunction of all transitions from x to y . These transitions constitute all possible words that lead from x to y without passing through any state.

If $x = y$, we also have to consider the empty word, since its run on A starts in x and ends in y . Thus, the corresponding regular expression is the disjunction of all transitions from x to y and ε .

In the case of a non-empty X , we make the following observation. If we pick an element $z \in X$, then any run σ from x to y either passes through z , or does not pass through z . If it does, we can split it into three parts.

- (i) The first part contains the prefix of σ which contains all states up to the first occurrence of z that is not the starting state.
- (ii) The second part contains that part of the remainder of σ which contains all further occurrences of z , though not the last state if that is z .
- (iii) The third part contains the remainder.

Parts (i) and (iii) can easily be expressed in terms of $X \setminus \{z\}$. Part (ii) can be further decomposed into runs from z to z that do not pass through z . Thus, part (ii) can also be expressed in terms of $X \setminus \{z\}$ with the help of the $*$ operator.

If σ does not pass through z , it is covered by the regular expression for paths from x to y restricted to $X \setminus \{z\}$.

In order to define R recursively, we need to pick an element $z \in X$ if $X \neq \emptyset$. For this purpose, we assume an ordering on Q . We will then pick $z \in X$ such that z is the smallest element in X w.r.t to the ordering.

Definition 4.5.12. Let $X \subseteq Q$. Let $x, y \in Q$. We define R recursively on $|X|$:

$$R_{x,y}^X := \begin{cases} \sum_{\substack{a \in \Sigma \\ (x,a,y) \in \delta}} a & \text{if } X = \emptyset \wedge x \neq y; \\ \sum_{\substack{a \in \Sigma \\ (x,a,y) \in \delta}} a + \varepsilon & \text{if } X = \emptyset \wedge x = y; \\ R_{x,z}^{X \setminus \{z\}} (R_{z,z}^{X \setminus \{z\}})^* R_{z,y}^{X \setminus \{z\}} + R_{x,y}^{X \setminus \{z\}} & \text{if } X \neq \emptyset \wedge z \text{ minimal in } X. \end{cases}$$

The formalization of R is more involved than its mathematical definition. We give some auxiliary definitions to keep the definition of R as compact and readable as possible. `nPlus` is \sum on regular expressions. `dfa_step_any` is the list of transitions from x to y . `R0` covers the case of $X = \emptyset$.

explain
measure

Definition `nPlus rs := foldr (@Plus char) (Void _) rs.`

Definition `dfa_step_any x y := enum ([pred a | dfa_step A x a == y]).`

Definition `R0 x y := let r := nPlus (map (@Atom _) (dfa_step_any x y)) in
if x == y then Plus r (Eps _) else r.`

Function `R (X: {set A}) (x y: A) {measure [fun X => #|X|] X} :=
match [pick z in X] with
| None => R0 x y
| Some z => let X' := X \ z in
Plus (Conc (R X' x z) (Conc (Star (R X' z z)) (R X' z y))) (R X' x y)
end.`

We now express $\mathcal{L}(A)$ using R . Based on the observation that every accepted word has a run from s to some state $f \in F$, we only have to combine the corresponding regular expressions $R_{s,f}^Q$ to form a regular expression for $\mathcal{L}(A)$. The goal of this chapter is to prove the following theorem.

Theorem 4.5.13. $\mathcal{L}(A)$ is recognizable by a regular expression, i.e.

$$\mathcal{L}\left(\sum_{f \in F} R_{s,f}^Q\right) = \mathcal{L}(A).$$

In order to prove this theorem, we will first define a predicate on words that corresponds to $\mathcal{L}(R_{x,y}^X)$. We call this predicate $L_{x,y}^X$ and define it such that it includes those words whose runs on A starting in x only pass through states in X and end in y .

Definition 4.5.14. Let $w \in \Sigma^*$. Let $X \subseteq Q$, and $x, y \in X$. Let σ be the run of w on A starting in x . We define $L_{x,y}^X$ such that

$$w \in L_{x,y}^X \iff \sigma_{|\sigma|-1} = y \wedge \forall i \in [1, |\sigma| - 2]. \sigma_i \in X.$$

The formalization of L requires some infrastructure. To check the second condition of L , we want to be able to state properties of all but the last items in a run. We define a function `belast` to remove the last element from a sequence. Note that, mathematically, runs include the starting state. In our formalization, this is not the case. Thus, we do not need to remove the first state from a run to retrieve all states the run passes through.

Definition `allbutlast xs := all p (belast xs).`

Definition `L (X: {set A}) (x y: A) :=
[pred w | (last x (dfa_run' A x w) == y)
&& allbutlast (mem X) (dfa_run' A x w)].`

We now prove properties of L that we will need for our proof of Theorem 4.5.13.

Lemma 4.5.15. L is monotone in X , i.e.

$$\forall X \subseteq Q, z \in X, x, y \in Q. L_{x,y}^X \subseteq L^{X \cup \{z\}} xy.$$

Proof. This follows directly from $X \subset X \cup \{z\}$. □

Lemma `L_monotone (X: {set A}) (x y z: A): {subset L^X x y <= L^(z | X) x y}.`

Lemma 4.5.16. The empty word is contained in $L_{x,y}^X$ if and only if $x = y$.

Proof. This follows immediately from the definition of L . □

Lemma `L_nil X x y: reflect (x = y) ([::] \in L^X x y).`

Next, we will prove that words whose run passes through a state z can be split into two words. The run of the first word will end in z , i.e. not pass through z .

Lemma 4.5.17. Let $w \in \Sigma^*$. Let $x, z \in Q$. Let σ be the run of w on A starting in x . Let $z \in \sigma_1 \dots \sigma_{|\sigma|-1}$. Then there exist $w_1, w_2 \in \Sigma^*$ such that

$$w = w_1 w_2 \wedge |w_2| < |w| \wedge z \notin \sigma_1 \dots \sigma_{|w_1|-1} \wedge \sigma_{|w_1|} = z.$$

Proof. Let i be the first occurrence of z in $\sigma_1 \dots \sigma_{|\sigma|-1}$ such that $\sigma_i = z$ and $i > 0$. Let $w_1 := w_0 \dots w_{i-1}$ and $w_2 := w_i \dots w_{|w|-1}$. The claim follows. □

Lemma $\text{run_split } x \ z \ w: z \ \backslash \text{in } \text{dfa_run}' \ A \times w \rightarrow$
 $\text{exists } w_1, \text{ exists } w_2,$
 $w = w_1 ++ w_2 \ \wedge$
 $\text{size } w_2 < \text{size } w \ \wedge$
 $z \ \backslash \text{notin } \text{belast } (\text{dfa_run}' \ A \times w_1) \ \wedge$
 $\text{last } x \ (\text{dfa_run}' \ A \times w_1) = z.$

We will make use of this fact in the next lemma, which splits words in L^X into two parts, the first of which is again in L^X . This will be quintessential later, when we split words in L^X into three parts that correspond to the recursive definition of R^X .

Lemma 4.5.18. *Let $X \subseteq Q$ and $x, y, z \in Q$. Let $w \in L_{x,y}^{X \cup \{z\}}$. We have that either*

$$w \in L_{x,y}^X$$

or there exist w_1 and w_2 such that

$$w = w_1 w_2 \ \wedge \ |w_2| < |w| \ \wedge \ w_1 \in L_{x,z}^X \ \wedge \ w_2 \in L_{z,y}^{X \cup \{z\}}.$$

Proof. We first eliminate the case of $z \in X$, which is solved trivially. Let σ be the run of w on A starting in x . We do a case distinction on $z \in \sigma_1 \dots \sigma_{|\sigma|-1}$.

1. For $z \notin \sigma_1 \dots \sigma_{|\sigma|-1}$ we can easily show $w \in L_{x,y}^X$.
2. For $z \in \sigma_1 \dots \sigma_{|\sigma|-1}$ we use Lemma 4.5.17 to split w in w_1 and w_2 . From $w \in L_{x,y}^{X \cup \{z\}}$ and $\sigma_{|w_1|} = z$ we immediately get $w_2 \in L_{z,y}^{X \cup \{z\}}$. We have $z \notin \sigma_1 \dots \sigma_{|\sigma|-1}$. We also have that $X = (X \cup \{z\}) \setminus \{z\}$ from $z \notin X$. Thus, we get $w_1 \in L_{x,z}^X$. The remainder of the claim follows directly from Lemma 4.5.17.

□

Before we show that L^X respects the defining equation of R^X , we have to show that we can combine words from $L_{x,z}^X$, $(L_{z,z}^X)^*$, and $L_{z,y}^X$ to form a word in $L_{x,y}^{X \cup \{z\}}$. We prove a general concatenation lemma for L^X .

Lemma 4.5.19. *Let $X \subseteq Q$, $x, y \in Q$, and $z \in X$. Let $w_1 \in L_{x,z}^X$ and $w_2 \in L_{z,x}^X$. Then we have*

$$w_1 w_2 \in L_{x,y}^X.$$

Proof. By $z \in X$, $w_1 \in L_{x,y}^X$, and $\sigma_{|w_1|} = z$ we get $\sigma_1, \dots, \sigma_{|w_1|} \in X$. We also have $\sigma_{|w_1|+1}, \dots, \sigma_{|\sigma|-2} \in X$ and $\sigma_{|\sigma|-1} = y$. Thus, $w_1 w_2 \in L_{x,y}^X$. □

Lemma $\text{L_cat } (X: \{\text{set } A\}) \times y \ z \ w_1 \ w_2:$
 $z \ \backslash \text{in } X \rightarrow$
 $w_1 \ \backslash \text{in } L^X \times z \rightarrow$
 $w_2 \ \backslash \text{in } L^X \times y \rightarrow$
 $w_1 ++ w_2 \ \backslash \text{in } L^X \times y.$

Lemma 4.5.20. *Let $n \in \mathbb{N}$. Let $w_0, \dots, w_{n-1} \in L_{z,z}^X$. We have that*

$$w_0 \dots w_{n-1} \in L_{z,z}^{X \cup \{z\}}.$$

Proof. By induction on n .

1. For $n = 0$ we have to prove $\varepsilon \in L_{z,z}^{X \cup \{z\}}$ which holds by Lemma 4.5.16.
2. For $n = n' + 1$ we have $w_0 \dots w_{n-2} \in L_{z,z}^{X \cup \{z\}}$ by inductive hypothesis. We also have $w_0, \dots, w_{n-1} \in L_{z,z}^X$ by assumption, and, thus, $w_{n-1} \in L_{z,z}^X$. By Lemma 4.5.19 we get $w_0 \dots w_{n-1} \in L_{z,z}^{X \cup \{z\}}$.

□

Lemma $L_flatten (X: \{set\ A\})\ z\ vv: \quad all\ (L^X\ z\ z)\ vv \rightarrow$
 $flatten\ vv\ \backslash in\ L^X(z\ |: X)\ z\ z.$

Finally, we can show that L^X respects the defining equation of R^X . With all the lemmas we have in place now, this can now be shown with relative ease.

Lemma 4.5.21. *Let $X \subseteq Q$, $x, y \in Q$, and $z \in X$. We have that*

$$L_{x,y}^{X \cup \{z\}} = L_{x,z}^X (L_{z,z}^X)^* L_{z,y}^X + L_{x,y}^X.$$

Proof. “ \Rightarrow ” By induction on $|w|$.

1. For $|w| = 0$ we get $w \in L_{x,y}^X$ by Lemma 4.5.16.
2. For $|w| > 0$ we get w_1 and w_2 such that $w = w_1 w_2$, $w_1 \in L_{x,z}^X$ and $w_2 \in L_{z,y}^{X \cup \{z\}}$. By inductive hypothesis we get

$$w_2 \in L_{z,z}^X (L_{z,z}^X)^* L_{z,y}^X \vee w_2 \in L_{z,y}^X$$

The latter gives us $w \in L_{x,y}^{X \cup \{z\}}$ by Lemma 4.5.19. With the former, we have w_3, w_4 , and w_5 such that $w_2 = w_3 w_4 w_5$, $w_3 \in L_{z,z}^X$, $w_4 \in (L_{z,z}^X)^*$ and $w_5 \in L_{z,y}^X$. We merge w_3 and w_4 such that $w_3 w_4 \in (L_{z,z}^X)^*$. This gives us $w_2 \in (L_{z,z}^X)^* L_{z,y}^X$. Thus, $w_1 w_2 \in L_{x,z}^X (L_{z,z}^X)^* L_{z,y}^X$.

“ \Leftarrow ” We have $w_1 \in L_{x,z}^X$, $w_2 \in (L_{z,z}^X)^*$, and $w_3 \in L_{z,y}^X$. By Lemma 4.5.20 we get $w_2 \in L_{z,z}^{X \cup \{z\}}$. Thus, $w_1 w_2 w_3 \in L_{x,y}^{X \cup \{z\}}$ by Lemma 4.5.19. □

Lemma $L_rec (X: \{set\ A\})\ x\ y\ z:$

$$L^X(z\ |: X)\ x\ y = i\ plus\ (conc\ (L^X\ x\ z)\ (conc\ (star\ (L^X\ z\ z))\ (L^X\ z\ y)))\ (L^X\ x\ y).$$

All that remains to complete the proof of Lemma 4.5.13 is a proof of $L_{x,y}^X = \mathcal{L}(R_{x,y}^X)$.

Lemma 4.5.22. *Let $X \subseteq Q$. Let $x, y \in Q$. We have that*

$$L_{x,y}^X = R_{x,y}^X.$$

Proof. By induction on $|X|$.

1. For $|X| = 0$, the claim follows immediately from the definitions of L and R .
2. For $|X| = n + 1$ for some $n \in \mathbb{N}$ we get $\exists z \in X$ and thus

$$R_{x,y}^X = R_{x,z}^{X \setminus \{z\}} (R_{z,z}^{X \setminus \{z\}})^* R_{z,y}^{X \setminus \{z\}} + R_{x,y}^{X \setminus \{z\}}.$$

By Lemma 4.5.21 we also know that

$$L_{x,y}^X = L_{x,z}^{X \setminus \{z\}} (L_{z,z}^{X \setminus \{z\}})^* L_{z,y}^{X \setminus \{z\}} + L_{x,y}^{X \setminus \{z\}}.$$

The claim follows by inductive hypothesis.

□

Lemma $L \cdot R^n (X: \{\text{set } A\}) \times y: \#|X| = n \rightarrow L^X \times y = R^X \times y$.

This concludes the proof of Theorem 4.5.13.

Lemma `dfa_to_regex`: **exists** `r`: `regular_expression` `char`, `dfa_lang` `A` `=i` `r`.

Chapter 5

Myhill-Nerode

In this chapter, we consider three additional characterizations of regular languages:

1. Myhill relations,
2. weak Nerode relations,
3. and Nerode relations.

We will show that these three characterizations can be used to characterize regular languages by proving them equivalent to the existence of a (deterministic) finite automaton.

5.1 Definitions

Before we can state the Myhill-Nerode theorem, we need a number of auxiliary definitions. We roughly follow [23].

Definition 5.1.1. *Let \equiv be an equivalence relation. The **equivalence class** of $u \in \Sigma^*$ w.r.t. \equiv is the set of all v such that $u \equiv v$. It is denoted by $[u]_{\equiv}$.*

Definition 5.1.2. *Let \equiv be an equivalence relation. \equiv is of **finite index** if and only if the set of $\{[u]_{\equiv} \mid u \in \Sigma^*\}$ is finite.*

Due to the lack of native support for quotient types in COQ, we formalize equivalence relations of finite index as surjective functions from Σ^* to a finite type X .

Definition 5.1.3. *Let X be finite. Let $f : \Sigma^* \mapsto X$ be surjective. Let $u, v \in \Sigma^*$. f is an **equivalence relation of finite index**. u and v are equivalent w.r.t. f if and only if $f(u) = f(v)$. $f(u)$ is the equivalence class of u w.r.t. f .*

Record Fin_Eq_Cls :=
 { fin_type : finType;
 fin_func :> word -> fin_type;
 fin_surjective : surjective fin_func }.

Definition 5.1.4. Let f be as above. Let $x \in X$. $w \in \Sigma^*$ is a **representative** of x if and only if $f(w) = x$. Since f is surjective, every w has a representative. We write $cr(x)$ to denote the **canonical representative** of x , which we obtain by constructive choice.

Definition $cr (f: Fin_Eq_Cls) x := xchoose (fin_surjective f x)$.

5.1.1 Myhill Relations

Definition 5.1.5. Let \equiv be an equivalence relation of finite index. \equiv is a **Myhillrelation** [23] on L if and only if

(i) \equiv is **right congruent**, i.e. for all $u, v \in \Sigma^*$ and $a \in \Sigma$,

$$u \equiv v \Rightarrow u \cdot a \equiv v \cdot a.$$

(ii) \equiv **refines** L , i.e. for all $u, v \in \Sigma^*$,

$$u \equiv v \Rightarrow (u \in L \iff v \in L).$$

Myhill relations are commonly referred to as “Myhill-Nerode relations”. In this thesis, it makes sense to split the concept of a Myhill relation from that of the Nerode relation. Apart from the Nerode relation, which can be seen as the coarsest Myhill relation, we also define weak Nerode relations that have no direction connection to Myhill relations. Thus, we strictly separate the characterizations.

Mathematically, Myhill relations are required to be of finite index. We only formalize equivalence relations of finite index. Thus, proving that a Myhill relation is of finite index mathematically corresponds to constructing a Myhill relation in our formalization.

Definition right_congruent $\{X\}$ ($f: word \rightarrow X$) :=
forall $u v a, f u = f v \rightarrow f (rcons u a) = f (rcons v a)$.

Definition refines $\{X\}$ ($f: word \rightarrow X$) :=
forall $u v, f u = f v \rightarrow u \setminus in L = (v \setminus in L)$.

Record Myhill_Rel :=
 { myhill_func :> Fin_Eq_Cls;
 myhill_congruent: right_congruent myhill_func;
 myhill_refines : refines myhill_func }.

Myhill relations correspond to the equivalence relations defined as the pairs of words (u, v) whose runs on a DFA A end in the same state. These relations are right congruent, refine $\mathcal{L}(A)$ and are of finite index as A has finitely many states. We will later give a formal proof of this.

5.1.2 Nerode Relations

Definition 5.1.6. Let $u, v \in \Sigma^*$. Let L be a language. We define the **Nerode relation** $\dot{=}_L$ on L such that

$$u \dot{=}_L v \iff \forall w \in \Sigma^*. uw \in L \iff vw \in L.$$

The Nerode relation given above is a propositional statement in CoQ. To proof that the Nerode relation is of finite index, we require an equivalence relation, i.e. a function f from words to a finite type, such that f is equivalent to $\dot{=}_L$.

Definition equiv_suffix {X} (f: word \rightarrow X) :=
forall u v, f u = f v \iff suffix_equal u v.

Record Nerode_Rel :=
 { nerode_func :> Fin_Eq_Cls;
 nerode_equiv: equiv_suffix nerode_func }.

Definition 5.1.7. Let L be a language and let \equiv be an equivalence relation. We say that \equiv is a **weak Nerode relation** on L if and only if

$$\forall u, v \in \Sigma^*. u \equiv v \implies u \dot{=}_L v.$$

Definition suffix_equal u v :=
forall w, u++w \in L = (v++w \in L).

Definition imply_suffix {X} (f: word \rightarrow X) :=
forall u v, f u = f v \implies suffix_equal u v.

Record Weak_Nerode_Rel :=
 { weak_nerode_func :> Fin_Eq_Cls;
 weak_nerode_imply: imply_suffix weak_nerode_func }.

It appears that the notion of a weak Nerode relation is not found in the literature. We will later prove them weaker than Myhill relations, in the sense that every Myhill relation is also a weak Nerode relation.


5.1.3 Myhill-Nerode Theorem

We can now move on to the statement of the Myhill-Nerode theorem [23].

Theorem 5.1.8. (Myhill-Nerode) Let L be a language. The following four statements are equivalent:

1. there exists a deterministic finite automaton that accepts L ;
2. there exists a Myhill relation on L ;
3. there exists a weak Nerode relation on L ;
4. the Nerode relation on L is of finite index.

Our proof of equivalence will have the following structure:



We will first show \textcircled{a} , \textcircled{b} , and \textcircled{d} . We will then give a proof of \textcircled{c} , which is the most challenging proof and formalization in this chapter.

5.2 Finite Automata, Myhill relations, and Nerode relations

First, we will show that, given a DFA A , we can construct a Myhill relation on $\mathcal{L}(A)$. We will then show that this also enables us to construct a weak Nerode relation $\mathcal{L}(A)$.

Conversely, knowing that the Nerode relation on language L is of finite index, we will show that we can construct a DFA that accepts L .

5.2.1 Finite Automata to Myhill relations

We assume we are given a DFA A . We will be using the states of A as equivalence classes. Our goal is to construct a Myhill relation, for which we will need an equivalence relation of finite index. Therefore, we first need to ensure that the mapping from words to equivalence classes is surjective. Thus, we consider the equivalent, connected automaton $A_c = (Q_c, s_c, F_c, \delta_c)$ (Definition 4.2.1), which has only reachable states. This enables us to construct a surjective function from words to the states of A_c .

Definition 5.2.1. Let $u \in \Sigma^*$. Let σ be the run of u on A_c . We define $f_M : \Sigma^* \mapsto Q_c$ such that $f_M(u)$ is the last state in σ , i.e.

$$f_M(u) := \sigma_{|\sigma|-1}.$$

Note that f_M is surjective (follows directly from Lemma 4.2.5) and, thus, an equivalence relation of finite index.

Definition $f_M := \text{fun } w \Rightarrow \text{last } (\text{dfa_s } A_c) (\text{dfa_run } A_c\ w).$

Lemma $f_M_surjective$: surjective f_M .

Definition $f_fin : \text{Fin_Eq_Cls} :=$
 $\{ | \text{ fin_func } := f_M;$
 $\text{ fin_surjective } := f_M_surjective \}.$

In order to show that f_M is a Myhill relation, we prove that it fulfills Definition 5.1.5.

Lemma 5.2.2. f_M is right congruent.

Proof. Let $u, v \in \Sigma^*$ such that $f_M(u) = f_M(v)$. Let $a \in \Sigma$. Since A is deterministic, we get $f_M(ua) = f_M(va)$. \square

Lemma 5.2.3. f_M refines $\mathcal{L}(A_c)$.

Proof. Let $u, v \in \Sigma^*$ such that $f_M(u) = f_M(v)$. By definition of f_M , the runs u and v on A end in the same state. Thus, either u and v are both accepted, or both not accepted. \square

Theorem 5.2.4. f_M is a Myhill relation on $\mathcal{L}(A)$.

Proof. By Lemma 4.2.3, we have $\mathcal{L}(A_c) = \mathcal{L}(A)$. Thus, it suffices to show that f_M is a Myhill relation on $\mathcal{L}(A_c)$. This follows with Lemma 5.2.2 and Lemma 5.2.3. \square

We only have extensional equality on $\mathcal{L}(A_c)$ and $\mathcal{L}(A)$ in COQ. Thus, we first show that f_M is a Myhill relation on $\mathcal{L}(A_c)$. Then, we show that we can get a Myhill relation on $\mathcal{L}(A)$ from a Myhill relation on $\mathcal{L}(A_c)$.

Definition `dfa_to_myhill' : Myhill_Rel (dfa_lang A.c) :=`
`{ | myhill_func := f_fin ;`
`myhill_congruent := f_M.right_congruent ;`
`myhill_refines := f_M.refines | }.`

Lemma `myhill_lang_eq L1 L2: L1 =i L2 -> Myhill_Rel L1 -> Myhill_Rel L2.`

Definition `dfa_to_myhill : Myhill_Rel (dfa_lang A) :=`
`myhill_lang_eq (dfa_connected_correct A) dfa_to_myhill'.`

This concludes the proof of step (a).

5.2.2 Myhill relations to weak Nerode relations

We show that, if there exists a Myhill relation, there also exists a weak Nerode relation. In fact, we will prove that any Myhill relation is a weak Nerode relation.

Theorem 5.2.5. Let f be a Myhill relation on a language L . Then f is a weak Nerode relation on L .

Proof. Let $u, v \in \Sigma^*$ such that $u =_f v$. We have to show that for all $w \in \Sigma^*$, $uw \in L \Leftrightarrow vw \in L$. By induction on w .

1. For $w = \varepsilon$, we get $u \in L \Leftrightarrow v \in L$ as f refines L .
2. For $w = aw'$, we get $ua =_f va$ by congruence of f and thus, by inductive hypothesis, $uaw' \in L \Leftrightarrow vaw' \in L$.

\square

Lemma myhill_suffix: imply_suffix L f.

Definition myhill_to_weak_nerode: Weak_Nerode_Rel L :=
 { | weak_nerode_func := f;
 weak_nerode_imply := myhill_suffix | }.

This concludes step (b) of Theorem 5.1.8.

5.2.3 Nerode relations to Finite Automata

We prove step (d) of Theorem 5.1.8. If the Nerode relation on a language L is of finite index, we can construct a DFA that accepts L . The construction is very straight-forward and uses the equivalence classes of the Nerode relation as the set of states for the automaton.

Definition 5.2.6. Let L be a language. Let X be a finite type. Let $f : \Sigma^* \mapsto X$ be the equivalence relation which proves that the Nerode relation on L is of finite index. We construct DFA A such that

$$\begin{aligned} s &:= f(\varepsilon) \\ F &:= \{x \mid x \in X \wedge cr(x) \in L\} \\ \delta &:= \{(x, a, f(cr(x)a)) \mid x \in X, a \in \Sigma\} \\ A &:= (X, s, F, \delta). \end{aligned}$$

Definition nerode_to_dfa :=
 { | dfa_s := f [::];
 dfa_fin := [pred x | cr f x \in L];
 dfa_step := [fun x a => f(rcons (cr f x) a)] | }.

In order to show that A accepts the language L , we first need to connect runs on A to the equivalence classes, i.e. the range of f . The following lemma gives a direct connection.

Lemma 5.2.7. Let $w \in \Sigma^*$. Let σ be the run of w on A starting in s . We have that the last state of σ is the equivalence class of w , i.e.

$$\sigma_{|\sigma|-1} = f(w).$$

Proof. We proceed by induction on w from right to left.

1. For $w = \varepsilon$ we have $s = f(\varepsilon)$.
2. For $w = w'a$ we know that the run of w' on A starting in s ends in $f(w')$. It remains to show that $(f(w'), a, f(w)) \in \delta$. We have $cr(f(w'))a =_f w$, i.e. $f(cr(f(w'))a) = f(w)$ by definition of f . Thus, it suffices to show $(f(w'), a, f(cr(f(w'))a)) \in \delta$, which holds by definition of δ .

□

Theorem 5.2.8. *A accepts L, i.e. $\mathcal{L}(A) = L$.*

Proof. Let $w \in \Sigma^*$. Let σ be the run of w on A starting in s . w is accepted if and only if $\sigma_{|\sigma|-1} \in F$, i.e. if and only if $cr(\sigma_{|\sigma|-1}) \in L$. We have $w =_f cr(\sigma_{|\sigma|-1})$ and therefore $w \in L \Leftrightarrow cr(\sigma_{|\sigma|-1}) \in L$. Thus w is accepted if and only if $w \in L$. \square

The resulting automaton is minimal, i.e. there exists no other automaton that accepts the same language and has less states than A .

This concludes step (d) of Theorem 5.1.8.

5.3 Minimizing Equivalence Classes

Finally, we prove that if there is a weak Nerode relation on a language L , the Nerode relation is of finite index. This proves step (c) of the Theorem 5.1.8 and thus completes its proof. For this purpose, we employ a table-filling algorithm [19] to find indistinguishable states under the Myhill-Nerode relation. This algorithm was originally intended to be used on automata. It identifies (un)distinguishable states based on their successors. We use the finite type X , i.e., the equivalence classes, instead of states.

For the remainder of this section, we assume we are given a language L and a weak Nerode relation f_0 .

We employ a fixed-point construction to find equivalence classes that are $\dot{=}_L$ -equivalent. In each step, we add those equivalence classes that are distinguishable based on equivalence classes that were distinguishable in the previous step. The initial set of distinguishable equivalence classes are distinguishable by the inclusion of their canonical representative in L . We denote this initial set $dist_0$.

$$dist_0 := \{(x, y) \in F \times F \mid cr(x) \in L \Leftrightarrow cr(y) \notin L\}.$$

Definition distinguishable := [fun x y => (cr f_0 x) \in L != ((cr f_0 y) \in L)].

Definition dist0 := [set x | distinguishable x.1 x.2].

To find more distinguishable equivalence classes, we have to identify equivalence classes that “lead” to distinguishable equivalence classes. In analogy to the minimization procedure on automata, we define successors of equivalence classes with respect to a single character. The intuition is that two states are distinguishable if they are succeeded by a pair of distinguishable states. Conversely, if a pair of states is not distinguishable, then their predecessors can not be distinguished by those states. Thus, two states are undistinguishable if none of their succeeding pairs of states are distinguishable.

Definition 5.3.1. Let $x, y \in X$ and $a \in \Sigma$. We define $succ_a$ and $psucc_a$. $succ_a(x) := f_0(cr(x) \cdot a)$ and $psucc_a(x, y) := (succ_a(x), succ_a(y))$.

Definition $\text{succ} := [\text{fun } x \ a \Rightarrow f_0 ((\text{cr } f_0 \ x) \ ++ \ [::a])]$.

Definition $\text{psucc} := [\text{fun } x \ y \Rightarrow [\text{fun } a \Rightarrow (\text{succ } x \ a, \text{succ } y \ a)]]$.

The fixed-point algorithm tries to extend the set of distinguishable equivalence classes by looking for a pair of equivalence classes that transitions to a pair of distinguishable equivalence classes. Given a set of pairs of equivalence classes dist , we define the set of pairs of distinguishable equivalence classes that have successors in dist as

$$\text{dist}_S(\text{dist}) := \{(x, y) \mid \exists a. \text{psucc}_a(x, y) \in \text{dist}\}.$$

Definition $\text{distS} \ (\text{dist} : \{\text{set } X * X\}) :=$
 $[\text{set } (x, y) \mid x \text{ in } X, y \text{ in } X \ \& \ [\text{exists } a, \text{psucc } x \ y \ a \ \backslash \text{in } \text{dist}]]$.

Definition 5.3.2. Let dist be a subset of $X \times X$. We define *one-step-dist* such that

$$\text{one-step-dist}(\text{dist}) := \text{dist}_0 \cup \text{dist} \cup \text{distinct}_S(\text{dist}).$$

Definition $\text{one_step_dist} \ \text{dist} := \text{dist}_0 \ :|: \ \text{dist} \ :|: \ (\text{distS} \ \text{dist})$.

Lemma 5.3.3. *one-step-dist* is monotone and has a fixed-point.

Proof. Monotonicity follows directly from the monotonicity of \cup . The number of sets in $X \times X$ is finite. Therefore, *one-step-dist* has a fixed point. We iterate *one-step-dist* $|X * X| + 1 = |X|^2 + 1$ times on the empty set. Since there can only ever be $|X * X|$ items in the result of *one-step-dist*, we will find the fixed point in no more than $|X * X| + 1$ steps.

Let **distinct** be the fixed point of *one-step-dist* and let it be denoted by $\not\cong$. We write **equiv** for the complement of **distinct** and denote it \cong . \square

Definition $\text{lfp} := \text{iter } \#|T|. + 1 \ F \ \text{set0}$.

Definition $\text{distinct} := \text{lfp} \ \text{one_step_dist}$.

We now show that \cong is equivalent to the Nerode relation. Formally, this means constructing a function that fulfills our definition of an equivalence relation of finite index. Then, we prove that this equivalence relation is equivalent to the the Nerode relation. First, we will prove that \cong is an equivalence relation. Then, we will prove it equivalent to the Nerode relation in two separate steps, since the two directions require different strategies.

Lemma 5.3.4. \cong is an equivalence relation.

Proof. We first state transitivity of \cong in terms of $\not\cong$:

$$\forall x, y, z \in X. \neg(x \not\cong y) \implies \neg(y \not\cong z) \implies \neg(x \not\cong z). \quad (*)$$

It suffices to show that **distinct** is anti-reflexive, symmetric and fulfills (*). Note that complementary transitivity, anti-reflexivity and symmetry are closed under union. We proceed by fixed-point induction.

1. For $\text{one-step-dist}(\text{dist}) = \emptyset$ we have anti-reflexivity, symmetry and (*) by the properties of \emptyset .
2. For $\text{one-step-dist}(\text{dist}) = \text{dist}'$ we have dist anti-reflexive, symmetric and (*). It remains to show that dist_0 and $\text{distinct}_S(\text{dist})$ are anti-reflexive, symmetric and fulfill (*).

dist_0 is anti-reflexive, symmetric and fulfills (*) by definition.

$\text{distinct}_S(\text{dist})$ can be seen as an intersection of a symmetric subset of $X \times X$ defined by psucc_a and dist , the latter of which is anti-reflexive, symmetric and fulfills (*). Thus, $\text{distinct}_S(\text{dist})$ is anti-reflexive, symmetric and fulfills (*).

Therefore, dist' is anti-reflexive, symmetric and fulfills (*).

□

Lemma `equiv_refl` x : $x \sim = x$.

Lemma `equiv_sym` x y : $x \sim = y \rightarrow y \sim = x$.

Lemma `equiv_trans` x y z : $x \sim = y \rightarrow y \sim = z \rightarrow x \sim = z$.

Lemma 5.3.5. Let $u, v \in \Sigma^*$. $f_0(u) \cong f_0(v) \implies u \dot{=}_L v$.

Proof. Let $w \in \Sigma^*$. We then show the contraposition of the claim:

$$uw \in L \not\Leftarrow vw \in L \implies f_0(u) \not\cong f_0(v).$$

By induction on w .

1. For $w = \varepsilon$ we have $u \in L \not\Leftarrow v \in L$ which gives us $(f_0(u), f_0(v)) \in \text{dist}_0$. Thus, $f_0(u) \not\cong f_0(v)$.
2. For $w = aw'$ we have $uaw \in L \not\Leftarrow vaw \in L$. We have to show $f_0(u) \not\cong f_0(v)$, i.e. $(f_0(u), f_0(v)) \in \text{distinct}$. By definition of distinct , it suffices to show $(f_0(u), f_0(v)) \in \text{one-step-dist}(\text{distinct})$.

For this, we prove $(f_0(u), f_0(v)) \in \text{distinct}_S(\text{distinct})$. By $uaw \in L \not\Leftarrow vaw \in L$ we have $(f_0(\text{cr}(u)a), f_0(\text{cr}(v)a)) \in \text{dist}_0$.

It remains to show that $f_0(\text{cr}(u)a) \not\cong f_0(\text{cr}(v)a)$ which we get by inductive hypothesis. For this, we need to show that $\text{cr}(u)aw \in L \not\Leftarrow \text{cr}(v)aw \in L$.

By the properties of f , we get $\text{cr}(u)aw \in L \Leftrightarrow uaw \in L$ and $\text{cr}(v)aw \in L \Leftrightarrow vaw \in L$. Thus, $\text{cr}(u)aw \in L \not\Leftarrow \text{cr}(v)aw$.

□

Lemma 5.3.6. Let $u, v \in \Sigma^*$. If $f_0(u) \not\cong f_0(v)$, then u and v are **not** equivalent wr.t. the Merode relation, i.e. $f_0(u) \not\cong f_0(v) \implies u \not\dot{=}_L v$.

Proof. We do a fixed-point induction.

1. For $\text{one-step-dist}(\text{dist}) = \emptyset$ we have $(f_0(u), f_0(v)) \in \emptyset$ and thus a contradiction.
2. For $\text{one-step-dist}(\text{dist}) = \text{dist}'$ we have $(f_0(u), f_0(v)) \in \text{dist}'$. We do a case distinction on dist' .
 - (a) $(f_0(u), f_0(v)) \in \text{dist}_0$. We have $u \in L \not\Rightarrow v \in L$. Thus, $u \not\equiv_L v$ as witnessed by $w = \varepsilon$.
 - (b) $(f_0(u), f_0(v)) \in \text{dist}$. By inductive hypothesis, $u \not\equiv_L v$.
 - (c) $(f_0(u), f_0(v)) \in \text{distinct}_S(\text{dist})$. We have $a \in \Sigma$ with $\text{psucc}_a(f_0(u), f_0(v)) \in \text{dist}$. By inductive hypothesis, we get $\text{cr}(u)a \not\equiv_L \text{cr}(v)a$ as witnessed by $w \in \Sigma^*$ such that $\text{cr}(u)aw \in L \not\Rightarrow \text{cr}(v)aw \in L$.
By the properties of f , we get $\text{cr}(u)aw \in L \Leftrightarrow uaw \in L$ and $\text{cr}(v)aw \in L \Leftrightarrow vaw \in L$. Thus, we have $u \not\equiv_L v$ as witnessed by aw .

□

Corollary 5.3.7. *Let $u, v \in \Sigma^*$. We have that*

$$f_0(u) \cong f_0(v) \iff u \dot{\equiv}_L v.$$

Proof. Follows immediately with Lemma 5.3.5 and Lemma 5.3.6. □

Lemma `equiv_suffix_equal u v: u ~=_f_0 v -> suffix_equal L u v.`

Lemma `distinct_not_suffix_equal u v:`

`u ~!=_f_0 v ->`

`exists w, u ++ w \in L != (v ++ w \in L).`

Lemma `equivP u v:`

`reflect (suffix_equal L u v)`

`(u ~=_f_0 v).`

Definition 5.3.8. *Let $w \in \Sigma^*$. We define*

$$f_{\min}(w) := \{x \mid x \in X, f_0(w) \cong x\}.$$

Note that the range of f_{\min} is finite (since X is finite) and does not contain the empty set (due to reflexivity of \cong).

Lemma 5.3.9. *f_{\min} is surjective.*

Proof. Let $s \in \text{range}(f_{\min})$. Since $s \neq \emptyset$, there exists $x \in X$ such that $x \in s$. We have $f_0(x) = f_0(\text{cr}(x))$ and therefore $f_0(x) \cong f_0(\text{cr}(x))$ by reflexivity of \cong . Thus, $f_0(\text{cr}(x)) = s$ since $f_{\min}(x) = f_{\min}(\text{cr}(x)) = s$. □

Lemma 5.3.10. *For all $u, v \in \Sigma^*$ we have*

$$f_{\min}(u) = f_{\min}(v) \iff f_0(u) \cong f_0(v).$$

Proof. “ \Rightarrow ” We have $f_{\min}(u) = f_{\min}(v)$ and thus $f_0(u) \cong f_0(v)$.

“ \Leftarrow ” We have $f_0(u) \cong f_0(v)$. Let $x \in X$. It suffices to show that $f_0(u) \cong x$ if and only if $f_0(v) \cong x$. This follows with symmetry and transitivity of \cong . \square

Lemma 5.3.11. *f_{\min} is equivalent to the Nerode relation, i.e. f_{\min} is surjective and for all $u, v \in \Sigma^*$ we have*

$$f_{\min}(u) = f_{\min}(v) \iff u \dot{=}_L v.$$

Proof. We have proven surjectivity in Lemma 5.3.9. By Lemma 5.3.10 we have $f_{\min}(u) = f_{\min}(v)$ if and only if $f_0(u) \cong f_0(v)$. By corollary 5.3.7 we have $f_0(u) \cong f_0(v)$ if and only if $u \dot{=}_L v$. Thus, $f_{\min}(u) = f_{\min}(v)$ if and only if $u \dot{=}_L v$. \square

The formalization of f_{\min} is slightly more involved than the mathematical construction. We first need to define the finite type of f_{\min} ’s range, which we do by enumerating all possible values of f_{\min} .

Definition `equiv_repr x := [set y | x \sim y].`

Definition `X_min := map equiv_repr (enum (fin_type f_0)).`

Definition `f_min w := SeqSub _ (equiv_repr_mem (f_0 w)).`

We then prove lemmas 5.3.9, 5.3.10 and Theorem 5.3.11 which are consequential and straight-forward.

Lemma `f_min_surjective: surjective f_min.`

Lemma `f_minP u v:`

$$\begin{aligned} &\text{reflect } (f_{\min} u = f_{\min} v) \\ &\quad (u \sim_{f_0} v). \end{aligned}$$

Lemma `f_min_correct: equiv_suffix L f_min.`

Definition `f_min_fin : Fin.Eq_Cls :=`

$$\{ | \text{fin_surjective} := f_{\min_surjective} | \}.$$

We can now state the result of this chapter.

Corollary 5.3.12. *The Nerode relation is of finite index.*

Proof. This follows directly from Lemma 5.3.4 and Lemma 5.3.11. \square

Definition `weak_nerode_to_nerode: Nerode_Rel L :=`

$$\{ | \text{nerode_func} := f_{\min_fin} ; \\ \text{nerode_equiv} := f_{\min_correct} | \}.$$

This concludes step \textcircled{C} of Theorem 5.1.8.

Chapter 6

Conclusion

Bibliography

- [1] *2nd Annual Symposium on Switching Circuit Theory and Logical Design, Detroit, Michigan, USA, October 17-20, 1961*. IEEE Computer Society, 1961.
- [2] Valentin M. Antimirov. Partial derivatives of regular expressions and finite automaton constructions. *Theor. Comput. Sci.*, 155(2):291–319, 1996.
- [3] Dean N. Arden. Delayed-logic and finite-state machines. In *SWCT (FOCS)* [1], pages 133–151.
- [4] Robert L. Ashenurst and Susan Graham, editors. *ACM Turing award lectures: the first twenty years: 1966-1985*. ACM Press/Addison-Wesley Publishing Co., New York, NY, USA, 1987.
- [5] Andrea Asperti. A compact proof of decidability for regular expression equivalence. In Beringer and Felty [8], pages 283–298.
- [6] Andrea Asperti, Claudio Sacerdoti Coen, and Enrico Tassi. Regular expressions, au point. *CoRR*, abs/1010.2604, 2010.
- [7] Andrea Asperti, Wilmer Ricciotti, Claudio Sacerdoti Coen, and Enrico Tassi. The matita interactive theorem prover. In Bjørner and Sofronie-Stokkermans [9], pages 64–69.
- [8] Lennart Beringer and Amy P. Felty, editors. *Interactive Theorem Proving - Third International Conference, ITP 2012, Princeton, NJ, USA, August 13-15, 2012. Proceedings*, volume 7406 of *Lecture Notes in Computer Science*. Springer, 2012.
- [9] Nikolaj Bjørner and Viorica Sofronie-Stokkermans, editors. *Automated Deduction - CADE-23 - 23rd International Conference on Automated Deduction, Wroclaw, Poland, July 31 - August 5, 2011. Proceedings*, volume 6803 of *Lecture Notes in Computer Science*. Springer, 2011.
- [10] Thomas Braibant and Damien Pous. Deciding kleene algebras in coq. *Logical Methods in Computer Science*, 8(1), 2012.

- [11] J. A. Brzozowski and E. J. McCluskey. Signal Flow Graph Techniques for Sequential Circuit State Diagrams. *IEEE Transactions on Electronic Computers*, EC-12:67–76, 1963.
- [12] Janusz A. Brzozowski. Derivatives of regular expressions. *J. ACM*, 11(4):481–494, 1964.
- [13] Thierry Coquand and Vincent Siles. A decision procedure for regular expression equivalence in type theory. In *CPP*, pages 119–134, 2011.
- [14] D.Z. Du and K.I. Ko. *Problem Solving in Automata, Languages, and Complexity*. Wiley, 2001.
- [15] Georges Gonthier. The four colour theorem: Engineering of a formal proof. In Kapur [21], page 333.
- [16] Georges Gonthier, Assia Mahboubi, and Enrico Tassi. A Small Scale Reflection Extension for the Coq system. Rapport de recherche RR-6455, INRIA, 2008.
- [17] Yo-Sub Han and Derick Wood. The generalization of generalized automata: Expression automata. In Michael Domaratzki, Alexander Okhotin, Kai Salomaa, and Sheng Yu, editors, *Implementation and Application of Automata*, volume 3317 of *Lecture Notes in Computer Science*, pages 156–166. Springer Berlin / Heidelberg, 2005. 10.1007/978-3-540-30500-2_15.
- [18] John E. Hopcroft, Rajeev Motwani, and Jeffrey D. Ullman. *Introduction to automata theory, languages, and computation - international edition (2. ed)*. Addison-Wesley, 2003.
- [19] D.A. Huffman. The synthesis of sequential switching circuits. *Journal of the Franklin Institute*, 257(3):161 – 190, 1954.
- [20] Wolfram Kahl and Timothy G. Griffin, editors. *Relational and Algebraic Methods in Computer Science - 13th International Conference, RAMiCS 2012, Cambridge, UK, September 17-20, 2012. Proceedings*, volume 7560 of *Lecture Notes in Computer Science*. Springer, 2012.
- [21] Deepak Kapur, editor. *Computer Mathematics, 8th Asian Symposium, ASCM 2007, Singapore, December 15-17, 2007. Revised and Invited Papers*, volume 5081 of *Lecture Notes in Computer Science*. Springer, 2008.
- [22] S. C. Kleene. Representation of events in nerve nets and finite automata. *Automata Studies*, 1965.
- [23] Dexter Kozen. *Automata and computability*. Undergraduate texts in computer science. Springer, 1997.

- [24] Alexander Krauss and Tobias Nipkow. Proof pearl: Regular expression equivalence and relation algebra. *J. Autom. Reasoning*, 49(1):95–106, 2012.
- [25] Peter Linz. *An introduction to formal languages and automata (4. ed.)*. Jones and Bartlett Publishers, 2006.
- [26] The Coq development team. *The Coq proof assistant reference manual*. LogiCal Project, 2004. Version 8.0.
- [27] Edward F. Moore. Gedanken-experiments on sequential machines. In Claude Shannon and John McCarthy, editors, *Automata Studies*, pages 129–153. Princeton University Press, Princeton, NJ, 1956.
- [28] Nelma Moreira, David Pereira, and Simão Melo de Sousa. Deciding regular expressions (in-)equivalence in coq. In Kahl and Griffin [20], pages 98–113.
- [29] Anil Nerode. Linear automaton transformations. *Proceedings of the American Mathematical Society*, 9(4):541–544, ao 1958.
- [30] M. O. Rabin and D. Scott. Finite automata and their decision problems. *IBM J. Res. Dev.*, 3(2):114–125, April 1959.
- [31] Marko C. J. D. van Eekelen, Herman Geuvers, Julien Schmaltz, and Freek Wiedijk, editors. *Interactive Theorem Proving - Second International Conference, ITP 2011, Berg en Dal, The Netherlands, August 22-25, 2011. Proceedings*, volume 6898 of *Lecture Notes in Computer Science*. Springer, 2011.
- [32] Chunhan Wu, Xingyuan Zhang, and Christian Urban. A formalisation of the myhill-nerode theorem based on regular expressions (proof pearl). In van Eekelen et al. [31], pages 341–356.